

지역 및 전역 최적화를 활용한 SmartSpice 회로설계

소개

SmartSpice의 최적화 기능을 통해 회로의 변수 및 파라미터를 최적화합니다. 회로 설계의 관점에서, 일단의 변수 또는 회로 파라미터의 제한 (일반적으로, 설계 파라미터에 대한 하한 및 상한) 대비 특정 목표를 달성하는 데 관심이 있습니다. 일련의 초기 값과 목표 값으로부터, 범용 최적화 엔진은 목표에 도달할 때까지 모수 공간에서 반복적인 작업을 수행합니다. 회로는 정상 상태, 주파수 및 시간 영역 등의 가능한 모든 분석을 활용하여 최적화할 수 있습니다.

최적화에는 성능 측정 최적화와 함수 최적화의 두 가지 유형이 있습니다.

성능 측정 최적화에서 전기적 특성 일부가 사양을 충족하도록 변수와 소자/모델 파라미터가 변경됩니다. 전기적 특성의 예로는 지연, 상승/하강 시간, 트립 포인트, 최대/최소 전류, 이득, S-파라미터 또는 SmartSpice의 .MEASURE 구문으로 계산할 수 있는 추가적인 회로 성능 측정이 있습니다.

함수 최적화에서 회로에서 정의한 일부 함수가 사양을 충족하도록 변수와 파라미터가 변경됩니다. 예를 들어, DC, AC, 과도 분석의 목표 응답과 계산된 곡선이 일치하도록 합니다.

지역 및 전역 방법

최적화 알고리즘은 지역 최적화 방법 또는 전역 최적화 방법을 고려할 수 있습니다. 지역 최적화 방법은 양질의 검색 방향을 이용하며, 일반적으로 지역 최소값에 근접할 때 강력하고 빠릅니다. 전역 최적화 방법은 지역 최소값에 갇히지 않아야 하므로, 속도가 느리며 전역 최소값을 찾기 위해 확률적 검색을 사용할 수도 있습니다.

모든 최적화 방법은 공통적으로 솔루션의 합리적인 추정치를 나타내는 시작점을 가집니다. 파라미터의 시작 세트는 이전의 경험, 문제에 대한 지식 또는 설계 공간의 범위를 고려한 합리적인 가정에서 얻을 수 있습니다. 지역 최적화는 반복적으로 일련의 지점을 생성한 다음, 솔루션이 지정된 설계 조건을 충족하거나 더 이상의 진행이 어려울 때 종료합니다.

전역 최적화는 모든 지역 최소값 중에서 결함이 최소인 함수를 찾으려고 시도합니다. 따라서 이 방법은 전역 최소값을 탐색할 때 지역 최소값을 예측해야 하므로, 지역 최적화보다 더 복잡합니다. 이는 지역 최적화에 비해 시작점에서 더 멀리 떨어진 설계 공간을 조사할 수도 있습니다.

이제 SmartSpice에서 사용할 수 있는 다양한 최적화 알고리즘이 있습니다. 지역 최적화에 널리 사용하는 Levenberg-Marquardt 방법은 솔루션에 근접하여 시작할 때 성능이 우수합니다. 전역 최적화 방법은 Hooke-Jeeves, 시뮬레이션 어닐링, 병렬 템퍼링, 유전 알고리즘, 미분 진화 및 하이브리드 방법 등이 있습니다. 보다 자세한 내용은 SmartSpice 설명서를 참조하십시오.

설정

전역 최적화 구문은 기존 .MODIF 최적화 섹션과 동일합니다. 최적화 전략을 선택하려면 OPTIMIZER 옵션을 다음과 같이 설정하십시오.

```
.OPTION OPTIMIZER=val
```

여기서 val은 전역 최적화 유형으로서, 다음과 같습니다.

- **LM** – Levenberg-Marquardt
- **HJ** – Hooke-Jeeves method
- **SA** – Simulated annealing
- **PT** – Parallel tempering
- **GA** – Genetic algorithm
- **DE** – Differential evolution
- **H** – Hybrid optimizer

기본값은 LM으로서 지역 최적화 툴, Levenberg-Marquardt를 나타냅니다.

최적화 알고리즘은 **Edit** → **Preferences...** → **Tools** → **Optimizer** 의 Preferences 대화 상자에서 설정할 수 있습니다. 이를 이용하여 중지 기준, 허용 오차, 반복, 시뮬레이션 수 및 기타 최적화 관련 파라미터 등의 다양한 관련 설정을 할 수 있습니다. 예를 들어, 그림 1은 시뮬레이션 어닐링 알고리즘의 최적화 설정을 나타냅니다.

시작 값은 [90nH, 0.5nH, 0.1pF, 25pF]입니다. 목표는 통과 대역에서 $|S_{21}| = -0.5\text{dB}$, 하부 및 상부 통과 대역에서 각각 -55dB (500MHz ~ 600MHz)와 -60dB (1.4GHz ~ 1.5GHz)였습니다. Levenberg-Marquardt 최적화를 수행한 후, 최종 요소 값은 [150.5nH, 0.908nH, 0.167pF, 27.7pF]입니다. 그림 3은 최적화 전후의 $|S_{21}|$ 결과를 나타냅니다.

설계 #1 - 밴드패스 필터

여기서, 통과 대역에서 0.5 dB의 equal-ripple 반응을 갖는 간단한 밴드패스 필터를 설계합니다. 중심 주파수는 1GHz이고 대역폭은 10%이며 임피던스는 50Ω입니다. 그림 2는 SmartSpice 넷리스트를 나타냅니다. L_13, L_2, c_13, c_2의 4가지 최적화 변수가 있습니다.




그림 1. 최적화 설정 대화 상자

```

* BAND PASS FILTER with z0=(50,50)
Vin IN 0 ac 1 dc 0
Vout OUT 0 ac 1 dc 0
L1 IN 1 L_13
C1 1 2 C_13
L2 2 0 L_2
C2 2 0 C_2
L3 2 3 L_13
C3 3 OUT C_13

.NET N=2 IN OUT 0 LIN 1001 500Meg 1.5G

.PARAM l_13=154.3nH l_2=0.666nH c_13=0.135pF c_2=39.33pF

.MEASURE NET passband MIN db(sm(out,in)) FROM=950Meg TO=1.05G
.MEASURE NET lower MAX db(sm(out,in)) FROM=500Meg TO=600Meg
.MEASURE NET upper MAX db(sm(out,in)) FROM=1400Meg TO=1500Meg

.MODIF
+ OPTIMIZE L_13=opt(10nH 200nH 90nH)
+           L_2 =opt(0.01nH 10nH 0.5nH)
+           C_13=opt(0.01pF 20pF 0.1pF)
+           C_2 =opt(5pF 300pF 25pF)
+ TARGETS passband=-0.5 lower=-55 upper=-60
+ OPTIONS AVG=1e-8 MEASOFF=1 NUMFUNC=300 NUMITER=300
+         IPLOFF=2 MARQUP=8000

.OPTIONS NOMOD iplot_one
.OPTION OPTIMIZER=LM
.IPLOT db(sm(out,in))
.END
    
```

그림 2. 밴드패스 필터의 SmartSpice 넷리스트




그림 3. Levenberg-Marquardt 최적화 전후의 |S21| 결과

설계 #2 - 광대역 증폭기 이득

여기서, 광대역 증폭기의 이득을 최적화합니다. 이 회로는 모든 저항 소자와 수많은 BJT 트랜지스터 파라미터를 포함하여 120개의 설계 파라미터를 사용합니다. 최종적으로 1Hz부터 10MHz까지, 최소 이득이 119이고 최대 이득이 120인 설계를 하려고 합니다. 지역 최적화 툴, Levenberg-Marquardt는 만족스러운 결과에 도달할 수 없습니다. 따라서, 우리는 특히 시뮬레이션 어닐링과 같은 옵션을 선택하여 전역 최적화를 활용합니다.

그림 4는 SmartSpice 넷리스트를 나타냅니다. 그림 5는 시뮬레이션 어닐링 최적화를 수행하기 전과 후의 결과를 나타냅니다. 이득은 122에서 균일하다가 10MHz에서 112로 떨어집니다. 대신에 시뮬레이션 어닐링에 이어 Levenberg-Marquardt 방법을 사용하는 하이브리드 최적화 전략을 선택한다면, 저주파수와 10MHz에서의 최종 결과는 각각 123과 114입니다. 하이브리드 최적화 툴을 사용하면, 두 번째 최적화 툴, Levenberg-Marquardt가 첫 번째 전략 (이 경우, 시뮬레이션 어닐링)에서 찾은 솔루션을 중심으로 지역 최적화를 추가합니다.

.OPTION OPTIMIZER=SA

EXAMPLE4: RCA3040 WIDEBAND AMPLIFIER

```
VIN 1 0 DC 0 SIN(0 0.1 50MEG 0.5NS) AC 1
VEE 3 0 -15
VCC 2 0 15
*
RS2 31 0 1K
RS1 30 1 1K
*
Q1 2 30 5 QNLA
Q2 2 31 6 QNLA
*
Q3 10 5 7 QNLB
Q4 11 6 7 QNLB
*
Q5 14 12 10 QNLC
Q6 15 12 11 QNLC
*
Q7 12 12 13 QNLD
Q8 13 13 0 QNLD
*
Q9 7 8 9 QNL
*
Q10 2 15 16 QNLE
Q11 2 14 17 QNLE
*
R1 5 3 ra
R2 6 3 rb
R3 9 3 rc
R4 8 3 rd
R5 8 0 re
R6 2 14 rf
R7 2 12 rg
R8 2 15 rh
R9 16 0 ri
R10 17 0 rj
*
.MODEL QNL NPN(BF=80 RB=100 CCS=2PF TF=0.3NS AF=1.0
```

```

+          KF=5.4E-16 TR=6NS  CJE=3PF  CJC=2PF  VA = 50 )

.MODEL  QNLA  NPN(BF=80      RB=100    CCS=2PF  TF=0.3NS  AF=1.0
+          NF=1.0      BR=2.38  NR=1.0    VJE=1.44  MJE=0.33
+          VJC=0.99    MJC=0.33  VJS=1.39  MJS=0.5   IS=92.5e-16
+          KF=5.4e-16 TR=6NS  CJE=3PF  CJC=2PF  VA=50)

.MODEL  QNLB  NPN(BF=80      RB=100    CCS=2PF  TF=0.3NS  AF=1.0
+          NF=1.0      BR=2.38  NR=1.0    VJE=1.44  MJE=0.33
+          VJC=0.99    MJC=0.33  VJS=1.39  MJS=0.5   IS=92.5e-16
+          KF=5.4e-16 TR=6NS  CJE=3PF  CJC=2PF  VA=50)

.MODEL  QNLC  NPN(BF=80      RB=100    CCS=2PF  TF=0.3NS  AF=1.0
+          NF=1.0      BR=2.38  NR=1.0    VJE=1.44  MJE=0.33
+          VJC=0.99    MJC=0.33  VJS=1.39  MJS=0.5   IS=92.5e-16
+          KF=5.4e-16 TR=6NS  CJE=3PF  CJC=2PF  VA=50)

.MODEL  QNLD  NPN(BF=80      RB=100    CCS=2PF  TF=0.3NS  AF=1.0
+          NF=1.0      BR=2.38  NR=1.0    VJE=1.44  MJE=0.33
+          VJC=0.99    MJC=0.33  VJS=1.39  MJS=0.5   IS=92.5e-16
+          KF=5.4e-16 TR=6NS  CJE=3PF  CJC=2PF  VA=50)

.MODEL  QNLE  NPN(BF=80      RB=100    CCS=2PF  TF=0.3NS  AF=1.0
+          NF=1.0      BR=2.38  NR=1.0    VJE=1.44  MJE=0.33
+          VJC=0.99    MJC=0.33  VJS=1.39  MJS=0.5   IS=92.5e-16
+          KF=5.4e-16 TR=6NS  CJE=3PF  CJC=2PF  VA=50)
*
**** Parameter labels
.PARAM  ra=4.8K  rb=4.8K  rc=811  rd=2.17K  re=820
+       rf=1.32K  rg=4.5K  rh=1.32K  ri=5.25K  rj=5.25K
*
**** MEASURE
.MEASURE AC gain_min MIN vm(16) FROM=1 TO=10e6
.MEASURE AC gain_max MAX vm(16) FROM=1 TO=10e6

.AC  DEC  5 1 10GHZ

.OPTIONS NOMOD iplot_one

**** Optimization specification
.MODIF
+   OPTIMIZE  ra=opt(1K  10K  2K)
+           rb=opt(1K  10K  2K)
+           rc=opt(100  2K  200)
+           rd=opt(500  5K  1K)
+           re=opt(100  2K  500)
+           rf=opt(500  5K  1K)
+           rg=opt(1K  10K  2K)
+           rh=opt(500  5K  12K)
+           ri=opt(1K  10K  2K)
+           rj=opt(1K  10K  2K)

```

```

+           QNL(BF)=OPT( 10  200  80 )
+           QNL(RB)=OPT( 10  500  100 )
+           QNL(CCS)=OPT( 0.5pF  10pF  2pF )
+           QNL(TF)=OPT( 0.1NS  1NS  0.3NS )
+           QNL(AF)=OPT( 0.2  5.0  1.0 )
+           QNL(KF)=OPT( 1e-18  1e-14  5.4e-16 )
+           QNL(TR)=OPT( 2ns  15ns  6ns )
+           QNL(CJE)=OPT( 0.5pF  10pF  3pF )
+           QNL(CJC)=OPT( 0.5pF  10pF  2pF )
+           QNL(VA)=OPT( 10  100  50 )

+           QNLA(BF)=OPT( 10  200  80 )
+           QNLA(RB)=OPT( 10  500  100 )
+           QNLA(CCS)=OPT( 0.5pF  10pF  2pF )
+           QNLA(TF)=OPT( 0.1NS  1NS  0.3NS )
+           QNLA(AF)=OPT( 0.2  5.0  1.0 )
+           QNLA(NF)=OPT( 0.1  3  1.0 )
+           QNLA(BR)=OPT( 0.2  3  2.38 )
+           QNLA(NR)=OPT( 0.5  3  1.0 )
+           QNLA(VJE)=OPT( 0.1  1.5  1.44 )
+           QNLA(MJE)=OPT( 0.1  1.5  0.33 )
+           QNLA(VJC)=OPT( 0.25  2  0.99 )
+           QNLA(MJC)=OPT( 0.1  1  0.33 )
+           QNLA(VJS)=OPT( 0.2  1.5  1.39 )
+           QNLA(MJS)=OPT( 0.1  2  0.5 )
+           QNLA(IS)=OPT( 1e-18  1e-14  92.51e-16 )
+           QNLA(KF)=OPT( 1e-18  1e-14  5.4e-16 )
+           QNLA(TR)=OPT( 2ns  15ns  6NS )
+           QNLA(CJE)=OPT( 0.5pF  10pF  3PF )
+           QNLA(CJC)=OPT( 0.5pF  10pF  2PF )
+           QNLA(VA)=OPT( 10  100  50 )

+           QNLB(BF)=OPT( 10  200  80 )
+           QNLB(RB)=OPT( 10  500  100 )
+           QNLB(CCS)=OPT( 0.5pF  10pF  2pF )
+           QNLB(TF)=OPT( 0.1NS  1NS  0.3NS )
+           QNLB(AF)=OPT( 0.2  5.0  1.0 )
+           QNLB(NF)=OPT( 0.1  3  1.0 )
+           QNLB(BR)=OPT( 0.2  3  2.38 )
+           QNLB(NR)=OPT( 0.5  3  1.0 )
+           QNLB(VJE)=OPT( 0.1  1.5  1.44 )
+           QNLB(MJE)=OPT( 0.1  1.5  0.33 )
+           QNLB(VJC)=OPT( 0.25  2  0.99 )
+           QNLB(MJC)=OPT( 0.1  1  0.33 )
+           QNLB(VJS)=OPT( 0.2  1.5  1.39 )
+           QNLB(MJS)=OPT( 0.1  2  0.5 )
+           QNLB(IS)=OPT( 1e-18  1e-14  92.51e-16 )
+           QNLB(KF)=OPT( 1e-18  1e-14  5.4e-16 )
+           QNLB(TR)=OPT( 2ns  15ns  6NS )
+           QNLB(CJE)=OPT( 0.5pF  10pF  3PF )
+           QNLB(CJC)=OPT( 0.5pF  10pF  2PF )

```

```

+           QNLB(VA)=OPT( 10 100 50 )

+           QNLC(BF)=OPT( 10 200 80 )
+           QNLC(RB)=OPT( 10 500 100 )
+           QNLC(CCS)=OPT( 0.5pF 10pF 2pF )
+           QNLC(TF)=OPT( 0.1NS 1NS 0.3NS )
+           QNLC(AF)=OPT( 0.2 5.0 1.0 )
+           QNLC(NF)=OPT( 0.1 3 1.0 )
+           QNLC(BR)=OPT( 0.2 3 2.38 )
+           QNLC(NR)=OPT( 0.5 3 1.0 )
+           QNLC(VJE)=OPT( 0.1 1.5 1.44 )
+           QNLC(MJE)=OPT( 0.1 1.5 0.33 )
+           QNLC(VJC)=OPT( 0.25 2 0.99 )
+           QNLC(MJC)=OPT( 0.1 1 0.33 )
+           QNLC(VJS)=OPT( 0.2 1.5 1.39 )
+           QNLC(MJS)=OPT( 0.1 2 0.5 )
+           QNLC(IS)=OPT( 1e-18 1e-14 92.51e-16 )
+           QNLC(KF)=OPT( 1e-18 1e-14 5.4e-16 )
+           QNLC(TR)=OPT( 2ns 15ns 6NS )
+           QNLC(CJE)=OPT( 0.5pF 10pF 3PF )
+           QNLC(CJC)=OPT( 0.5pF 10pF 2PF )
+           QNLC(VA)=OPT( 10 100 50 )

+           QNLD(BF)=OPT( 10 200 80 )
+           QNLD(RB)=OPT( 10 500 100 )
+           QNLD(CCS)=OPT( 0.5pF 10pF 2pF )
+           QNLD(TF)=OPT( 0.1NS 1NS 0.3NS )
+           QNLD(AF)=OPT( 0.2 5.0 1.0 )
+           QNLD(NF)=OPT( 0.1 3 1.0 )
+           QNLD(BR)=OPT( 0.2 3 2.38 )
+           QNLD(NR)=OPT( 0.5 3 1.0 )
+           QNLD(VJE)=OPT( 0.1 1.5 1.44 )
+           QNLD(MJE)=OPT( 0.1 1.5 0.33 )
+           QNLD(VJC)=OPT( 0.25 2 0.99 )
+           QNLD(MJC)=OPT( 0.1 1 0.33 )
+           QNLD(VJS)=OPT( 0.2 1.5 1.39 )
+           QNLD(MJS)=OPT( 0.1 2 0.5 )
+           QNLD(IS)=OPT( 1e-18 1e-14 92.51e-16 )
+           QNLD(KF)=OPT( 1e-18 1e-14 5.4e-16 )
+           QNLD(TR)=OPT( 2ns 15ns 6NS )
+           QNLD(CJE)=OPT( 0.5pF 10pF 3PF )
+           QNLD(CJC)=OPT( 0.5pF 10pF 2PF )
+           QNLD(VA)=OPT( 10 100 50 )

+           QNLE(BF)=OPT( 10 200 80 )
+           QNLE(RB)=OPT( 10 500 100 )
+           QNLE(CCS)=OPT( 0.5pF 10pF 2pF )
+           QNLE(TF)=OPT( 0.1NS 1NS 0.3NS )
+           QNLE(AF)=OPT( 0.2 5.0 1.0 )
+           QNLE(NF)=OPT( 0.1 3 1.0 )
+           QNLE(BR)=OPT( 0.2 3 2.38 )

```

```

+          QNLE(NR)=OPT( 0.5  3  1.0 )
+          QNLE(VJE)=OPT( 0.1  1.5  1.44 )
+          QNLE(MJE)=OPT( 0.1  1.5  0.33 )
+          QNLE(VJC)=OPT( 0.25  2  0.99 )
+          QNLE(MJC)=OPT( 0.1  1  0.33 )
+          QNLE(VJS)=OPT( 0.2  1.5  1.39 )
+          QNLE(MJS)=OPT( 0.1  2  0.5 )
+          QNLE(IS)=OPT( 1e-18  1e-14  92.51e-16 )
+          QNLE(KF)=OPT( 1e-18  1e-14  5.4e-16 )
+          QNLE(TR)=OPT( 2ns  15ns  6NS )
+          QNLE(CJE)=OPT( 0.5pF  10pF  3PF )
+          QNLE(CJC)=OPT( 0.5pF  10pF  2PF )
+          QNLE(VA)=OPT( 10  100  50 )

+      TARGETS  gain_min=119 gain_max=121
+      OPTIONS  AVG=1e-6 NUMFUNC=500 NUMITER=300
+              MEASOFF=1 IPLOFF=1 MARQUP=5000
+
+      ***** Select the optimizer
+      .OPTION OPTIMIZER=SA
+      .I PLOT VM(16)
+      .END
    
```

그림 4. 120개의 설계 변수가 있는 광대역 증폭기의 SmartSpice 넷리스트




그림 5. 시뮬레이션 어닐링 최적화 전후의 증폭기 이득 플롯

설계 #3 - 디지털 인버터

여기서, 출력 전압 1.95V로 그림 6의 디지털 인버터의 공급 전력을 최소화합니다.

설계 변수는 전력 공급기 전압이며, 목표는 직류원에서 공급하는 계산 전력을 위한 작은 값과 출력 전압 1.95V입니다. 그림 7은 넷리스트를 나타냅니다. 공급 전압 (V_SUPPLY) 시작점 2.5V에서 Levenberg-Marquardt 최적화로 지역 최소값을 얻습니다.

따라서, 다음과 같은 옵션을 선택하여 하이브리드 최적화 툴을 사용합니다.

.OPTION OPTIMIZER=H




그림 6. 디지털 인버터

Preferences 대화상자의 최적화 섹션에서 병렬 템퍼링과 Levenberg-Marquardt를 각각 첫 번째 및 두 번째 최적화 툴로 선택하면, 목표에 맞는 솔루션을 찾을 수 있습니다. 그림 8은 최적화 전후의 출력 전압을 보여주는 과도 결과를 나타냅니다.

```
* Digital inverter

***** SOME PARAMETERS
.PARAM INP_FREQ      = 850MegHz
.PARAM INP_PERIOD    = 1/INP_FREQ
.PARAM NO_PERIODS    = 4
.PARAM TMEAS_START   = '(NO_PERIODS-1)*INP_PERIOD'
.PARAM TMEAS_STOP    = NO_PERIODS*INP_PERIOD
.PARAM T1             = 'TMEAS_STOP - 1*INP_PERIOD/4'
.PARAM T2             = 'TMEAS_STOP - 0.75*INP_PERIOD/4'

***** OPTIMIZATION PARAMETERS
.PARAM V_SUPPLY      = 2V
.PARAM WP            = 1000u
.PARAM LMIN          = 0.35u

***** SUPPLY VOLTAGES
VDS (VDD 0) DC V_SUPPLY

***** INPUT SIGNAL
VSIG (IN 0) PULSE(V_SUPPLY 0
+          INP_PERIOD/2
+          INP_PERIOD/1000 INP_PERIOD/1000
+          INP_PERIOD/2 INP_PERIOD)

***** OUTPUT STAGE
MP (OUT IN VDD VDD) pm W=WP L=LMIN
MN (OUT IN 0 0) nm W=WP/3 L=LMIN

***** Models
.MODEL pm PMOS ( level=3 tox=.02e-6
+ phi=0.576 gamma=0 vto=0 alpha=0 kappa=0 is=0)
.model nm NMOS ( level=3 tox=.02e-6
+ phi=0.576 gamma=0 vto=0 alpha=0 kappa=0 is=0)

CL (OUT 0) 10pF
```

```

***** Analysis statement
.TRAN INP_PERIOD/250 NO_PERIODS*INP_PERIOD CALLV SAVEV

.OPTIONS iplot_one nomod
.IPLOT v(OUT)

**** Measure statements
.MEASURE TRAN VHIGH MAX v(OUT) FROM=T1 TO=T2
.MEASURE TRAN PSUPPLY MAX '(v(VDD)*i(VDS))' FROM=T1 TO=T2

.MODIF
+ OPTIMIZE V_SUPPLY=opt(1 5 2.5)
+ TARGETS PSUPPLY=1e-6 VHIGH=1.95
* VLOW=0.001
+ OPTIONS AVG=1e-6 MEASOFF=1 NUMFUNC=50 NUMITER=50
+ IPLOFF=2

***** Select the optimizer
.OPTION OPTIMIZER=H

.END
    
```

그림 7. 디지털 인버터의 SmartSpice 넷리스트




그림 8. 최적화 전후의 디지털 인버터 출력 전압 플롯

결론

SmartSpice 최적화 기능을 통해 주어진 사양을 충족하거나 개선하기 위해 설계를 개선할 수 있습니다. 일련의 초기 값과 지정 목표에서 최적의 솔루션을 달성할 때까지 최적화 톨은 모수 공간에 대해 반복적으로 수행합니다. 지역 및 전역 최적화를 모두 사용할 수 있습니다. 지역 최적화는 시작 점이 지역 최소값에 가까울 때 유용하며, 일반적으로 빠릅니다. 전역 최적화는 광범위한 파라미터 검색을 통해 전역 솔루션에 도달할 수 있습니다. 정상 상태, 주파수 및 시간 영역 등의 가능한 모든 분석을 사용하여 회로를 최적화할 수 있습니다. SmartSpice에 전역 최적화 기능이 추가되어, 수많은 파라미터를 활용하여 설계를 최적화할 수 있습니다.